

Основы GNUPLLOT

Компьютерная графика

Юдинцев В. В.

Кафедра теоретической механики


22 ноября 2024 г.



САМАРСКИЙ УНИВЕРСИТЕТ
SAMARA UNIVERSITY

- **свободная** программа для создания двухмерных и трёхмерных графиков в командном интерактивном режиме или в режиме выполнения команд из файла
- результат работы программы – изображение на экране или графический файл (.eps, .png, .pdf, ...)
- работает на всех основных платформах: Linux, Microsoft Windows, macOS
- первая версия программы вышла в 1986 году
- используется в качестве системы вывода изображений в различных свободных математических пакетах: GNU Octave, Maxima


<http://www.gnuplot.info/>



gnuplot homepage

[FAQ](#)
[Documentation](#)
[Demos](#)
[Download](#)

[Contributed scripts](#)
[External Links](#)
[Tutorials and guides](#)
[Books](#)



Gnuplot is a portable command-line driven graphing utility for Linux, OS/2, MS Windows, OSX, VMS, and many other platforms. The source code is copyrighted but freely distributed (i.e., you don't have to pay for it). It was originally created to allow scientists and students to visualize mathematical functions and data interactively, but has grown to support many non-interactive uses such as web scripting. It is also used as a plotting engine by third-party applications like Octave. Gnuplot has been supported and under active development since 1986.

Gnuplot supports many different types of 2D and 3D plots

Here is a [Gallery of demos](#).

Gnuplot supports many different types of output

interactive screen display:	cross-platform (Qt, wxWidgets, x11) or system-specific (MS Windows, OS/2)
direct output to file:	postscript (including eps), pdf, png, gif, jpeg, LaTeX, metafont, emf, svg, ...
mouseable web display	HTML5, svg

<https://sourceforge.net/projects/gnuplot/files/gnuplot/5.2.2/>

[Home](#) / [Browse](#) / [Graphics](#) / [Graphics](#) / [gnuplot](#) / [Files](#)



gnuplot

A portable, multi-platform, command-line driven graphing utility

Brought to you by: [broeker](#), [cgaylord](#), [lhecking](#), [sfeam](#)

[Summary](#)

[Files](#)

[Reviews](#)

[Support](#)

[Tickets ▼](#)

[gnuplot-main](#)

[Mailing Lists](#)

[Discussion](#)

[News](#)

[Code](#)

The interactive file manager requires Javascript. Please enable it or use [sftp](#) or [scp](#).

You may still *browse* the files here.



Download Latest Version
gnuplot-5.2.2.tar.gz (5.3 MB)

[Get Updates](#)




[Home](#) / [gnuplot](#) / 5.2.2

Name	Modified	Size	Downloads / Week
------	----------	------	------------------

↶ [Parent folder](#)

gp522-win64-mingw.exe	2017-11-18	26.5 MB	3,146
gp522-win32-mingw.exe	2017-11-18	26.9 MB	202
gp522-win32-mingw.7z	2017-11-18	25.2 MB	101
gp522-win64-mingw.7z	2017-11-17	24.8 MB	130



Philipp K. Janert
Gnuplot in Action
Second Edition
2016. 400 с.

Информация

en Страница проекта

<http://www.gnuplot.info/>

en Примеры

http://gnuplot.sourceforge.net/demo_5.2/

en Руководство пользователя

http://www.gnuplot.info/docs_5.2/Gnuplot_5.2.pdf

ru Нечаев А. Н. Краткое введение в GNUPLOT

<http://www.phys.vsu.ru/~meremianin/pdfs/gnuplot-gdoc.pdf>

ru Построение научных и инженерных графиков с помощью GNUPLOT

https://www.ibm.com/developerworks/ru/library/l-GnuPlot_01/

Спообы работы: диалоговый режим

vadim@bpc: ~

Файл Правка Вид Поиск Терминал Справка

vadim@bpc:~\$ gnuplot

G N U P L O T

Version 5.0 patchlevel 5 last modified 2016-10-02

Copyright (C) 1986-1993, 1998, 2004, 2007-2016
Thomas Williams, Colin Kelley and many others

gnuplot home: http://www.gnuplot.info
faq, bugs, etc: type "help FAQ"
immediate help: type "help" (plot window: hit 'h')


Terminal type set to 'qt'

gnuplot> set xrange [0:5]

gnuplot> plot sin(x**2)*x

gnuplot> █

Диалоговый режим




Файл-сценарий plot.gp

В текстовом редакторе (Блокнот, atom, sublimetext, ...) создаётся файл с инструкциями (командами) для gnuplot

```
1 # Способ вывода
2 set term pngcairo size 1200,800 font 'Times,26'
3 # Файл с результатом
4 set output 'plot1.png'
5 # Подписи осей
6 set xlabel "t, с"
7 set ylabel "y, м"
8 set title "Движение точки"
9 # Диапазон для горизонтальной оси
10 set xrange [0:2]
11 plot sin(x) title 'Перемещение'
```

Для выполнения скрипта в терминале вызывается программа gnuplot с параметром – именем файла: **gnuplot plot.gp**

Результат работы программы – plot1.png



Способы и параметры вывода

set terminal

Команда `set terminal` определяет способ вывода результата работы программы `gnuplot`:

- на экран
- в растровый файл `png`
- в векторный файл `pdf`, `eps`, `svg`, `epslatex`

Формат png: set terminal pngcairo

pngcairo – терминал для вывода в растровый графический формат .png.

- Размер изображения 640x480 точек шрифт 12 точек

```
1 | set terminal pngcairo font 'Arial,12'
```

- Размер изображения 1200x900 точек шрифт 26

```
1 | set terminal pngcairo size 1200,900 \  
2 |     font 'Times,26'
```

- Базовая толщина линий

```
1 | set terminal pngcairo size 1200,900 \  
2 |     font 'Times,26' linewidth 2
```

Формат **svg**: **set terminal svg**

pngcairo – терминал для вывода в векторный графический формат **.svg**.

- Размер изображения 640x480 точек шрифт 12 точек

```
1 | set terminal svg font 'Arial,12'
```

- Размер изображения 1200x900 точек шрифт 26

```
1 | set terminal svg size 1200,900 \  
2 |     font 'Times,26'
```

- Базовая толщина линий

```
1 | set terminal svg size 1200,900 \  
2 |     font 'Times,26' linewidth 2
```

Формат pdf

pdfcairo – терминал для вывода в векторный графический формат **.pdf**.

- Размер изображения 5х3 дюйма шрифт 12*(1/72) мм

```
1 | set terminal pdfcairo font 'Arial,12'
```

- Размер изображения 6х5 дюйма шрифт 26*(1/72) мм

```
1 | set terminal pdfcairo size 6,5 \  
2 |     font 'Times,26'
```

- Базовая толщина линий 2


```
1 | set terminal pdfcairo size 6,5 \  
2 |     font 'Times,26' linewidth 2
```

Графики функций

Функции

```
1 set term pngcairo size 1200,800 font "Times,26" \  
2 linewidth 2  
3 # Диапазон x  
4 set xrange [-5:5]  
5 # Диапазон y  
6 set yrange [-1.5:1.5]  
7 # Подписи осей  
8 set xlabel 'x'  
9 set ylabel 'f'  
10 # Файл вывода  
11 set output 'functions.png'  
12  
13 plot sin(x) with linespoints,\  
14      cos(x) with linespoints
```


Функции



Линии без маркеров

```
1 set term pngcairo size 1200,800 font "Times,26" \  
2 linewidth 2  
3 # Диапазон x  
4 set xrange [-5:5]  
5 # Диапазон y  
6 set yrange [-1.5:1.5]  
7 # Подписи осей  
8 set xlabel 'x'  
9 set ylabel 'f'  
10 # Файл вывода  
11 set output 'functions.png'  
12  
13 plot sin(x) with lines ,\  
14      cos(x) with lines
```


Линии без маркеров



Линии, маркеры, линии с маркерами

```
1 set term pngcairo size 1200,800 font "Times,26"  
  linewidth 2  
2  
3 set output 'line_style.png'  
4  
5 set xrange [-5:5]  
6 set yrange [-1.5:1.5]  
7 set xlabel 'x'  
8 set ylabel 'f'  
9  
10 plot sin(x) with lines , \  
11      cos(x) with points , \  
12      0.5*x with linespoints
```


Стили графика



Подпись данных (легенда)

```
1 set term pngcairo size 1200,800 font "Times,26" \  
2 linewidth 2  
3  
4 set key top, left  
5  
6 set xrange [-5:5]  
7 set yrange [-1.5:1.5]  
8 set xlabel 'x'  
9 set ylabel 'f'  
10  
11 set output 'functions.png'  
12  
13 plot sin(x) title 'Синус' with lines,\  
14      cos(x) title 'Косинус' with lines
```


Подпись данных (легенда)



Тип линии

```
1 set term pngcairo size 1200,800 font "Times,26" \
2   linewidth 3
3
4 set output 'line_style_line.png'
5
6 set xrange [-5:5]
7 set yrange [-1.5:1.5]
8 set xlabel 'x'
9 set ylabel 'f'
10
11 plot sin(x) with lines , \
12      cos(x) with lines dt '-', \
13      0.5*x with lines dt '_'
```

Тип линии



Создание стиля линии

Стиль линии – это сочетание её свойств:

- толщина;
- тип (сплошная, пунктирная, ...);
- цвет;
- типа маркера.

Красная линия толщиной 2, без маркера

```
1 | set style line 1 lw 2 lc rgb 'red' pt 1
```

Зелёная и синяя с маркерами типа 2 и 3

```
1 | set style line 2 lw 2 lc rgb 'green' pt 2  
2 | set style line 3 lw 2 lc rgb 'blue' pt 3
```

Тип маркера


·	0 : no symbol	△	8 : Up Triangle, empty
+	1 : Cross	▲	9 : Up Triangle, filled
×	2 : Saltire	▽	10 : Down Triangle, empty
✱	3 : Cross and Saltire	▼	11 : Down Triangle, filled
□	4 : Square, empty	◇	12 : Diamond, empty
■	5 : Square, filled	◆	13 : Diamond, filled
○	6 : Circle, empty	⬡	14 : Pentagon, empty
●	7 : Circle, filled	⬢	15 : Pentagon, filled

Пример

Свойство линии **pointinterval** регулирует 'частоту' маркеров

```
1 set term pngcairo size 1200,800 font "Times,26" linewidth 2
2
3 set style line 1 lw 2 lc rgb 'red' pt 0
4 set style line 2 lw 2 lc rgb 'green' pt 4 ps 2
5 set style line 3 lw 2 lc rgb 'blue' pt 10 ps 2 pointinterval 4
6
7 set output 'custom_lines_style.png'
8
9 set xrange [-5:5]
10 set yrange [-1.5:1.5]
11 set xlabel 'x'
12 set ylabel 'f'
13
14 plot sin(x) t '{/Symbol f}' with lines ls 1 ,\
15      cos(x) t 'x' with linespoints ls 2 ,\
16      0.5*x t '{/Symbol w}' with linespoints ls 3
```

Пример




Монохромные графики

Для подготовки графиков к монохромной печати можно определить собственные стили линий

```
1 set term pngcairo size 1200,800 font "Times,26" linewidth 2
2
3 set style line 1 lw 2 lc -1 pt 1
4 set style line 2 lw 2 lc -1 pt 2 dt (10,4)
5 set style line 3 lw 2 lc -1 pt 3 dt (10,4,4,4)
6
7 set output 'custom_lines_style_dashed.png'
8
9 set xrange [-5:5]
10 set yrange [-1.5:1.5]
11 set xlabel 'x'
12 set ylabel 'f'
13
14 plot sin(x) t '{/Symbol f}' with lines ls 1 ,\
15      cos(x) t 'x' with lines ls 2 ,\
16      0.5*x t '{/Symbol w}' with lines ls 3
```

Монохромные графики




multiplot

Несколько графиков


```
1 set term pngcairo size 1100,800 font "Times,18" linewidth 2
2 set output 'multiplot1.png'
3 set multiplot layout 2,2
4 set xrange [-5:5]
5 set yrange [-1.5:1.5]
6 set grid
7
8 set xlabel 'x'
9 set title 'Синус'
10 set ylabel 'sin(x)'
11 plot sin(x) t '' with lines lt -1
12
13 set title 'Косинус'
14 set ylabel 'cos(x)'
15 plot cos(x) t '' with lines ls 1
16
17 set title 'Тангенс'
18 set ylabel 'tan(x)'
19 plot tan(x) t '' with lines ls 2
20
21 set title 'Кубическая парабола'
22 set ylabel 'x^3'
23 plot x**3 t '' with lines dt '_'
```

Несколько графиков


Синус




Косинус



Тангенс



Кубическая парабола



Файлы с данными

Графики по табличным данным


Данные в столбац разделены пробелами

```
1 # Температура в Самаре (март)
2 # Минимум, среднее, максимум
3 # откл. от нормы, осадки мм
4 1 -22.0 -16.0 -10.3 -8.4 0.0
5 2 -21.7 -15.0 -7.2 -7.6 0.0
6 3 -19.6 -14.3 -9.5 -7.2 0.0
7 4 -14.7 -11.1 -6.7 -4.2 6.0
8 5 -9.2 -3.9 -0.6 +2.8 11.0
9 6 -14.3 -11.3 -8.0 -4.9 0.3
10 7 -19.7 -12.9 -7.3 -6.7 0.0
11 8 -13.0 -10.5 -7.4 -4.6 1.3
12 ...
13 31 -10.3 -5.9 -1.6 -7.5 0.0
```

График по табличным данным

```
1 set term pngcairo size 1200,800 font "Times,26" linewidth 2
2
3 set output '04-2018.png'
4 set xrange [1:31]
5
6 set xlabel 'День'
7 set ylabel 'Температура оC'
8 set title 'Темература в марте 2018 года'
9
10 set key bottom right
11
12 plot './temp.txt' using 1:2 t 'мин' with linespoints ps 2, \
13      './temp.txt' using 1:3 t 'сред' with linespoints ps 2, \
14      './temp.txt' using 1:4 t 'макс' with linespoints ps 2
```


Температура в марте 2018 года в г. Самара



Заливка кривых

```
1 set term pngcairo size 1200,800 font "Times,26" linewidth 2
2
3 set output '04-2018-filled.png'
4 set xrange [1:31]
5
6 set xlabel 'День'
7 set ylabel 'Температура оC'
8 set title 'Температура в марте 2018 года в г. Самара'
9
10 set key bottom right
11
12 set grid
13
14 set style fill transparent solid 0.25
15
16 plot './temp.txt' using 1:3:2 t '' with filledc lc rgb 'blue', \
17      './temp.txt' using 1:3:4 t '' with filledc lc rgb 'red', \
18      './temp.txt' using 1:3 t 'среднее' with lines lc rgb '#2222CC'
```


Температура в марте 2018 года в г. Самара




Формат файла

Разделителем столбцов по умолчанию является пробел. Для изменения разделителя используется команда

```
set datafile separator ','
```

```
1 set term pngcairo size 1200,800 font "Times,22" \
2   linewidth 2
3 # Разделитель столбцов в файле типа csv
4 set datafile separator ', '
5
6 set output 'default_data.png'
7
8 plot 'motion.csv' u 1:2 t '{/Symbol f}' ,\
9       'motion.csv' u 1:3 t 'x' ,\
10      'motion.csv' u 1:4 t '{/Symbol w}'
```


График



Тип маркера

```
1 set term pngcairo size 1200,800 font "Times,26" linewidth 2
2
3 set datafile separator ','
4 set output 'markers.png'
5
6 plot 'motion.csv' u 1:2 t '{/Symbol f}' with linespoints pt 4 ,\
7      'motion.csv' u 1:3 t 'x' with linespoints pt 6 ,\
8      'motion.csv' u 1:4 t '{/Symbol w}' with linespoints pt 10
```


Тип маркера



Вычисляемые значения

```
1 set term pngcairo size 1200,800 font "Times,26" linewidth 2
2
3 set datafile separator ','
4 set output 'expression.png'
5 set xlabel 't, c'
6 set ylabel 'Угол поворота, градус'
7
8 plot 'motion.csv' u 1:($2*180/pi) t '' with linespoints pt 4
```

Вычисляемые значения




Подписи оси x из файла

1	Меркурий	57910006	0.3871
2	Венера	108199995	0.7232
3	Земля	149599951	1.0000
4	Марс	227939920	1.5236
5	Юпитер	778330257	5.2028
6	Сатурн	1429400028	9.5549
7	Уран	2870989228	19.1913
8	Нептун	4504299579	30.1093

Подписи оси x из файла

```
1 set term pngcairo size 1200,800 \  
2     font "Times,24" linewidth 3  
3  
4 set output 'planets.png'  
5 set xrange [*:*]  
6 plot './planets.txt' using 3:xticlabels(1) \  
7     title '' with linespoints
```


Подписи оси x из файла



Логарифмическая шкала y

```
1 set term pngcairo size 1200,700 font "Times,24"  
   linewidth 3  
2  
3 set logscale y  
4  
5 set output 'planets_log.png'  
6 set xrange [*:*]  
7 plot './planets.txt' using 3:xticlabels(1) \  
8     title '' with linespoints ps 2 pt 7
```

Логарифмическая шкала у



Несколько блоков данных в файле

Файл blocks.txt. Блоки данных разделяются двумя пустыми строками:


```
1 # Первый блок (индекс 0)
2 # X      Y
3     1      2
4     2      3
5
6
7 # Второй блок (индекс 1)
8 # X      Y
9     3      2
10    4      1
```

Несколько блоков данных в файле

Данные блока адресуются при помощи директивы **index**

```
1 set term pngcairo size 1200,800 font "Times,18" \  
2 linewidth 2  
3 set output 'blocks.png'  
4  
5 plot 'blocks.txt' index 0 u 1:2 t 'B1' \  
6 with linespoints pt 4 ,\  
7 'blocks.txt' index 1 u 1:2 t 'B2' \  
8 with linespoints pt 6
```


Несколько блоков данных в файле



Две оси y

```
1 set term pngcairo size 1200,800 font "Times,24" linewidth 2
2
3 set ytics nomirror
4 set y2tics
5
6 set ylabel 'Скорость шарика, м/с'
7 set y2label 'Угол (градус), угловая скорость (градус/с)'
8
9 set key top left
10 set grid
11
12 set datafile separator ','
13
14 set output 'secondary.png'
15
16 plot 'motion.csv' u 1:($2*180/pi) t '{/Symbol f}' \
17         axes x1y2 with lines lw 2,\
18         'motion.csv' u 1:($2*180/pi) t '{/Symbol w}' \
19         axes x1y2 with lines lw 2,\
20         'motion.csv' u 1:5 t 'x' axes x1y1 with lines lw 2
```


Две оси y



Задание

Задание

- Установить [gnuplot](#)
- Экспортировать результаты интегрирования движения механизма (КР) в текстовый файл (время, положение шарика, угол поворота пластины, скорость шарика, угловая скорость пластины).
- Построить графики зависимостей от времени всех кинематических параметров системы (4 графика на отдельных рисунках в отдельных файлах).
- Построить графики зависимостей от времени всех кинематических параметров системы (на одном рисунке).
- Построить графики зависимостей от времени всех кинематических параметров системы (4 графика на одной странице), используя директиву [multiplot](#).